Program for Design of Power Transformer

Anil Kumar Malik

Power transformers are being used in power supplies and every electronic instrument So, one can design them without having to do a lot of calculations and referring to the tables for SWG and other data necessary to design, provided one has a microcomputer available. This article describes the design of a power transformer by computer programming in BASIC

Following points have been taken into consideration while making the program

1 The efficiency of transformer has been taken as 85 per cent

2 It can design a transformer with maximum of five secondary windings

3 Design is based on high-quality CRGO stampings and current density at 200 amps/cm^2 of enamelled copper wire

4 It will display "OUT OF RANGE" if current in any of the windings exceeds 16 6 amps

One can modify the data on maximum current capacity of the conductor, turns/cm^2 and SWG depending upon one's own requirement and specification This data can be changed in the DATA statements

After running the program the computer will ask you the following questions

1. Primary voltage in volts?
2. Frequency in Hz?
3. Wave-shape (Sine/Square)?
4. If you have more than one Sec winding?
5. Sec voltage in volts?
6. Sec current in amps?

After inputting this information, it will display the results in the following format

1 Primary turns and guage
2 First Sec turns and guage
3 Second Sec turns and guage
4 Third Sec. turns and guage
5 Fourth Sec turns and guage
6 Fifth Sec. turns and guage
7. Core area
8. Tongue width
9 Window area

With the results in (7), (8) and (9) one can select the type of core required for the transformer. One example is given in the program with the following specifications to make things clear:

1. Primary voltage in volts? 230
2. Frequency in Hz? 50
3. Wave-shape (Sine/Square)? Sine
4. If you have more than one Sec. voltages, Y/N? N
5. Sec. voltage in volts? 7.5
6. Sec. current in amps? 0.5

This program will run successfully on any microcomputer based on standard BASIC language. For personal computers the grammar of some of the statements will have to be changed or modified. The language of the BASIC statements

Anil Kumar Malik is scientist/engineer in Electronics Test and Development Centre, SAS Nagar

PROGRAM LISTING

```
5 REM PROGRAM TO DESIGN POWER TRANSFORMER
10 FOR M=1 TO 50
15 PRINT "*";
20 NEXT M
25 DIM A(40),B(40),W(40)
30 DATA 16.6,8.7,10,13.638,10.4,11,10.961,12.8,12,8.579,16.1
32 DATA 13,6.487,21.5,14,5.254,26.8,15,4.151,35.2,16,3.178
34 DATA 45.4,17,2.335,60.8,18,1.662,87.4,19,1.313,106,20
36 DATA 1.0377,137,21,0.7945,176,22,0.5838,224,23,0.4906
38 DATA 286,24,0.4054,341,25,0.3284,415,26,0.2726,504,27
40 DATA 0.2219,609,28,0.1874,711,29,0.1558,881,30,0.1364
41 DATA 997,31,0.1182,1137,32,0.1013,1308,33,0.0858,1608,34
42 DATA 0.0715,1902,35,0.0586,2286,36,0.0469,2800,37,0.0365
43 DATA 3507,38,0.0274,4838,39,0.0233,5595,40,0.0197,6543
44 DATA 41,0.0162,7755,42,0.0131,9337,43,0.0104,11457,44
45 DATA 0.0079,14392,45,0.0059,20223,46,0.0041,27546,47,0.0026
46 DATA 39706,48,0.0015,62134,49,0.0010,81242,50
50 FOR N=0 TO 40
55 READ A(N),B(N),W(N)
60 NEXT N
65 PRINT
70 INPUT "PRIMARY VOLTAGE IN VOLTS ",V(1)
75 INPUT "FREQUENCY IN HZ",F
80 INPUT "WAVE-SHAPE(SINE/SQUARE)";W$
85 IF W$="SINE" THEN 100
90 K=4
95 GOTO 105
100 K=4.44
105 INPUT "IF YOU HAVE MORE THAN ONE SEC. VOLTAGES,Y/N",A$
110 IF A$="Y" THEN 130
115 INPUT "SEC. VOLTAGE IN VOLTS",V(2)
120 INPUT "SEC. CURRENT IN AMPS ",I(2)
125 GOTO 150
130 PRINT "ASSIGN 0 VALUE FOR THE BALANCE VOLTAGES & CURRENTS"
135 INPUT "SEC. VOLTAGES IN VOLTS",V(2),V(3),V(4),V(5),V(6)
140 INPUT "SEC. CURRENTS IN AMPS ",I(2),I(3),I(4),I(5),I(6)
150 REM P2 IS OUTPUT POWER
155 P2=V(2)*I(2)+V(3)*I(3)+V(4)*I(4)+V(5)*I(5)+V(6)*I(6)
160 REM C1 IS X-SECTIONAL AREA
165 C1=1.152*SQR(P2)
170 REM T IS TURNS PER VOLT
180 T=1/(K*10^(-4)*F*C1*1.3)
185 REM I(1) IS PRIMARY CURRENT
190 I(1)=P2/(V(1)*.85)
200 REM T(1) IS PRIMARY TURNS
205 T(1)=V(1)*T
210 REM T(2),T(3)-----ARE SEC. TURNS
215 FOR P=1 TO 6
220 FOR D=1 TO 40
225 IF I(P)<16.6 THEN 590
230 IF I(P)=A(D) THEN 260
235 IF I(P)<A(D) THEN 280
240 IF I(P)=0 THEN 300
245 NEXT D
250 NEXT P
255 GOTO 330
260 Q(P)=B(D)
265 S(P)=W(D)
270 S$(P)="S.W.G"
275 GOTO 250
280 Q(P)=B(D-1)
285 S(P)=W(D-1)
290 S$(P)="S.W.G"
295 GOTO 250
300 Q(P)=10
305 S(P)=0
310 S$(P)="(NOT APPLICABLE)"
320 GOTO 250
330 REM Q(1),Q(2)-----ARE PRIMARY & SEC. TURNS/SQ.CM
340 REM S(1),S(2)-----ARE S.W.G FOR PRIMARY & SEC. WINDINGS
350 FOR E=2 TO 6
360 T(E)=T*V(E)*1.03
370 NEXT E
380 REM W1 IS WINDOW AREA
390 W1=1.3*(T(1)/Q(1)+T(2)/Q(2)+T(3)/Q(3)+T(4)/Q(4)+T(5)/Q(5))
395 W1=W1+1.3*(T(6)/Q(6))
400 REM C2 IS GROSS CORE AREA
410 C2=C1/.9
420 REM T7 IS TONGUE WIDTH
430 T7=SQR(C2)
433 T7=T7*100
434 T7=INT(T7)/100
435 C1=C1*100
436 C1=INT(C1)/100
437 W1=W1*100
438 W1=INT(W1)/100
440 FOR F=1 TO 6
450 T(F)=INT(T(F))
460 NEXT F
470 FOR M=1 TO 50
480 PRINT "*";
485 NEXT M
490 PRINT
495 PRINT
500 PRINT "   PRIMARY TURNS & GAUGE:";T(1),S(1);S$(1)
510 PRINT
520 PRINT "FIRST SEC. TURNS & GAUGE:";T(2),S(2);S$(2)
525 PRINT "SECOND SEC TURNS & GAUGE:";T(3),S(3);S$(3)
530 PRINT "THIRD SEC. TURNS & GAUGE:";T(4),S(4);S$(4)
535 PRINT "FOURTH SEC TURNS & GAUGE:";T(5),S(5);S$(5)
540 PRINT "FIFTH SEC. TURNS & GAUGE:";T(6),S(6);S$(6)
545 PRINT
550 PRINT "CORE AREA:",C1;"SQ.CM"
560 PRINT "TONGUE WIDTH:",T7;"CM"
570 PRINT "WINDOW AREA:",W1;"SQ.CM"
580 GOTO 600
590 PRINT "DESIGN OUT OF RANGE"
600 END
```

```
Primary Turns & Gauge : 3572      41 S.W.G
First Sec. Turns & Gauge: 119     23 S.W.G.
Second Sec. Turns & Gauge: 0      0 (Not Applicable)
Third Sec. Turns & Gauge:  0      0 (Not Applicable)
Fourth Sec. Turns & Gauge: 0      0  (Not Applicable)
Fifth Sec. Turns & Gauge:  0      0 (Not Applicable)
Core Area:          2.23 Sq. cm
Tongue Width:       1.57 cm
Window Area:        1.4 Sq. cm
```

accepted by the personal computers (such as the Sinclair ZX Spectrum) is somewhat different from the standard BASIC language. The statements accepted by the different personal computers may be the same but the grammar used in these statements differs. So, one has to modify the statements for the program depending upon the language of the computer.

For example, the statement 25 DIM A(40), B(40), W(40) (see program listing) may not be acceptable to the Sinclair ZX Spectrum PC if its memory is less, since this program requires 4k-byte memory. The statement 85 IF W$ = "SINE" THEN 100 which may not be acceptable to the PC as it is, may be acceptable with a slight change in the grammar as:

85 IF W$ = "SINE" GO TO 100 or

85 IF W$ = "SINE" THEN GO TO 100 □